﻿/*

  The contents of this file are subject to the Mozilla Public License Version
  1.1 (the "License"); you may not use this file except in compliance with
  the License. You may obtain a copy of the License at 
  
           http://www.mozilla.org/MPL/ 
  
  Software distributed under the License is distributed on an "AS IS" basis,
  WITHOUT WARRANTY OF ANY KIND, either express or implied. See the License
  for the specific language governing rights and limitations under the License. 
  
  The Original Code is Jutsu Opensource Framework.
  
  The Initial Developer of the Original Code is 
  ALCARAZ Marc <ekameleon@gmail.com>.
  Portions created by the Initial Developer are Copyright (C) 2012-2013
  the Initial Developer. All Rights Reserved.
  
  Contributor(s) :
  
  Alternatively, the contents of this file may be used under the terms of
  either the GNU General Public License Version 2 or later (the "GPL"), or
  the GNU Lesser General Public License Version 2.1 or later (the "LGPL"),
  in which case the provisions of the GPL or the LGPL are applicable instead
  of those above. If you wish to allow use of your version of this file only
  under the terms of either the GPL or the LGPL, and not to allow others to
  use your version of this file under the terms of the MPL, indicate your
  decision by deleting the provisions above and replace them with the notice
  and other provisions required by the LGPL or the GPL. If you do not delete
  the provisions above, a recipient may use your version of this file under
  the terms of any one of the MPL, the GPL or the LGPL.
  
*/

package jutsu.display
{
    import graphics.desktop.defaultDisplay;
    import graphics.desktop.descriptor;
    import graphics.display.Background;
    
    import lunas.components.buttons.FrameLabelButton;
    
    import system.hack;
    
    import flash.display.DisplayObject;
    
    /**
     * The header of the application.
     */
    public class Header extends Background
    {
        use namespace hack ;
        
        /**
         * Creates a new Header instance.
         */
        public function Header()
        {
            //
        }
        
        /**
         * The back button reference.
         */
        public var backButton:FrameLabelButton ;
        
        /**
         * The home button reference.
         */
        public var homeButton:FrameLabelButton ;
        
        /**
         * The logo of the header.
         */
        public var logo:DisplayObject ;
        
        /**
         * Initialize the canvas.
         * @param init An object that contains properties with which to populate the newly instance. If init is not an object, it is ignored.
         */
        public override function initialize( init:Object = null ):void 
        {
            _w = defaultDisplay.width  ;
            _h = int( defaultDisplay.height * 84 / descriptor.height ) ;
            super.initialize( init ) ;
        }
        
        /**
         * Invoked when the orientation of the stage is changed.
         */
        public function orientationChange( beforeOrientation:String , afterOrientation:String ):void
        {
            w = defaultDisplay.width ;
        }
        
        /**
         * Invoked when the orientation of the stage begins changing.
         */
        public function orientationChanging( beforeOrientation:String , afterOrientation:String ):void
        {
            w = defaultDisplay.width ;
        }
        
        /**
         * Invoked when the stage is resized.
         */
        public function resize():void
        {
            w = defaultDisplay.width ;
        }
        
        /**
         * Invoked when the view is updated.
         */
        protected override function viewChanged():void
        {
            if( logo )
            {
                logo.height = int( h * 75 / 84 ) ; // logo.height=75 and header.height=84
                logo.scaleX = logo.scaleY ;
                
                logo.y = ( h - logo.height  ) / 2 ;
            }
            
            const spacing:uint = int( w * 16 / descriptor.width ) ; 
            
            if( homeButton )
            {
                homeButton.height = int( h * 60 / 84 ) ; 
                homeButton.scaleX = homeButton.scaleY ;
                
                homeButton.x = w - homeButton.width - spacing ;
                homeButton.y = ( h - homeButton.height  ) / 2 ;
            }
            
            if( backButton )
            {
                backButton.height = int( h * 60 / 84 ) ; 
                backButton.scaleX = backButton.scaleY ;
                
                backButton.x = homeButton.x - backButton.width - 2 * spacing ;
                backButton.y = ( h - backButton.height  ) / 2 ;
            }
        }
    }
}
